#include <asm.h>
#include <memlayout.h>
#include <mmu.h>

#define REALLOC(x) (x + KERNBASE)

# Each non-boot CPU ("AP") is started up in response to a STARTUP
# IPI from the boot CPU.  Section B.4.2 of the Multi-Processor
# Specification says that the AP will start in real mode with CS:IP
# set to XY00:0000, where XY is an 8-bit value sent with the
# STARTUP. Thus this code must start at a 4096-byte boundary.
#
# Because this code sets DS to zero, it must sit
# at an address in the low 2^16 bytes.
#
# Bootothers (in main.c) sends the STARTUPs, one at a time.
# It puts this code (start) at 0x7000.
# It puts the correct %esp in start-4,
# and the place to jump to in start-8.
#
# This code is identical to bootasm.S except:
#   - it does not need to enable A20
#   - it uses the address at start-4 for the %esp
#   - it jumps to the address at start-8 instead of calling bootmain

.set PROT_MODE_CSEG,		0x8						# kernel code segment selector
.set PROT_MODE_DSEG,		0x10					# kernel data segment selector
.set CR0_PE_ON,				0x1						# protected mode enable flag

.code16                       # Assemble for 16-bit mode
.globl start
start:
	cli                       # Disable interrupts

	# Set up the important data segment registers (DS, ES, SS).
	xorw %ax,%ax             # Segment number zero
	movw %ax,%ds             # -> Data Segment
	movw %ax,%es             # -> Extra Segment
	movw %ax,%ss             # -> Stack Segment

	# Switch from real to protected mode, using a bootstrap GDT
	# and segment translation that makes virtual addresses 
	# identical to physical addresses, so that the 
	# effective memory map does not change during the switch.
	lgdt gdtdesc
	movl %cr0, %eax
	orl  $CR0_PE_ON, %eax
	movl %eax, %cr0

	# This ljmp is how you load the CS (Code Segment) register.
	# SEG_ASM produces segment descriptors with the 32-bit mode
	# flag set (the D flag), so addresses and word operands will
	# default to 32 bits after this jump.
	ljmp $PROT_MODE_CSEG, $start32

.code32                       # Assemble for 32-bit mode
start32:
	# Set up the protected-mode data segment registers
	# load temperate gdt
	lgdt __gdtdesc
	movl $PROT_MODE_DSEG, %eax
	movw %ax, %ds
	movw %ax, %es
	movw %ax, %ss
	movw %ax, %fs
	movw %ax, %gs

	ljmp $KERNEL_CS, $REALLOC(relocated)

relocated:

  # Set up the stack pointer and call into C.
	movl $0x0, %ebp
	movl $REALLOC(start)-4, %esp
	call  *(REALLOC(start)-8)

spin:
	jmp  spin

.data

.align PGSIZE
	.globl bootstack
bootstack:
	.space KSTACKSIZE
	.globl bootstacktop
bootstacktop:

.align 4
__gdt:
	SEG_NULL
	SEG_ASM(STA_X | STA_R, - KERNBASE, 0xFFFFFFFF)		# code segment
	SEG_ASM(STA_W, - KERNBASE, 0xFFFFFFFF)				# data segment
__gdtdesc:
	.word 0x17											# sizeof(__gdt) - 1
	.long __gdt

# Bootstrap GDT
.p2align 2                                # force 4 byte alignment
gdt:
  SEG_NULLASM                             # null seg
  SEG_ASM(STA_X|STA_R, 0x0, 0xffffffff)   # code seg
  SEG_ASM(STA_W, 0x0, 0xffffffff)         # data seg

gdtdesc:
  .word   (gdtdesc - gdt - 1)             # sizeof(gdt) - 1
  .long   gdt                             # address gdt
